

In 10 Schritten zur SAS GTL - Grafik

Marlis Kemena
 Siemens Healthcare Diagnostics Products
 GmbH HC LD QT MAR STA
 Emil-von-Behring-Str. 76
 35041 Marburg
 marlis.kemena@siemens-healthineers.com

Zusammenfassung

Der Beitrag beinhaltet eine Kurz-Einführung in SAS Graph Template Language (GTL) und vermittelt einen ersten Eindruck der Arbeitsweise und Möglichkeiten. Dabei werden die grundlegende Struktur sowie die Verwendung verschiedener Elemente und Plot Statements exemplarisch aufgezeigt. Die Grafik-Ausgabe erfolgt im PDF-Format.

Anhand eines Programmier-Beispiels ist zu erkennen, wie sich die daraus resultierende GTL-Grafik aus verschiedenen Bausteinen zusammensetzt:

- Step 1: ODS, Proc Template, Proc SGRRender
- Step 2: Title, Axes
- Step 3: Symbols
- Step 4: Regression Line, Confidence Limits
- Step 5: Line (Slope/Intercept), Connection of Data Points
- Step 6: Legend
- Step 7: Range
- Step 8: Reference Line, Label
- Step 9: Axis Scaling, Border
- Step 10: Summary

Die Darstellung in diesem Beitrag beschränkt sich auf die Schritte 1 und 10, da die Inhalte von Step 2 – 4 und 6 – 9 in Step 10 zusammengefasst sind.

Im Programmier-Beispiel geht es um Langzeit-Stabilitätsmessungen anhand fiktiver Daten, denen eine lineare Regressionsfunktion angepasst wird. Ein aus fachlicher Sicht relevanter Bereich mit seiner oberen und unteren Grenze erscheint in der Grafik blau hervorgehoben. Von Interesse ist der Punkt, wo die obere Grenze des 0,95-Konfidenzintervalls diesen blauen horizontalen Bereich erstmals schneidet. Der Schnittpunkt wird mit einer grünen senkrechten Linie markiert und beschriftet.

Schlüsselwörter: SAS GTL, ODS, PDF

1 Step 1: ODS, Proc Template, Proc SGRRender

1.1 SAS Programm: Step 1

Das Programm enthält ODS Statements (ODS = Output Delivery System) und die grundlegende Struktur von SAS GTL (Graph Template Language):

M. Kemena

```
LIBNAME KSFE          "D:\KSFE_GTL\Daten";
%LET Result_Path    = D:\KSFE_GTL\PDF;
OPTIONS    NODATE    NONUMBER    ORIENTATION=portrait    PAPERSIZE='A4'
           NOCENTER    TOPMARGIN=0.7CM    BOTTOMMARGIN=1.0CM
           LEFTMARGIN=2.0CM    RIGHTMARGIN=1.0CM;
GOPTIONS    RESET=ALL    DEVICE=winprtc    ROTATE=PORTRAIT;

ODS LISTING CLOSE;
ODS PDF FILE = "&Result_Path.\Step_01.pdf" NOTOC STYLE=PRINTER;
ODS LAYOUT START;
ODS GRAPHICS ON / BORDER=OFF          WIDTH=17.9cm    HEIGHT=11.9cm;
ODS REGION          X=0.0cm    Y=0.5cm    WIDTH=18.0cm    HEIGHT=12.0cm;

/* Grafik - Vorlage */
/*****/
PROC TEMPLATE;
    DEFINE STATGRAPH work.KSFE_GLT_Plot;
        BEGINGRAPH;
            ENTRYTITLE "ODS, Proc Template, Proc SGRender";

            LAYOUT OVERLAY;

                SCATTERPLOT    X = Hours    Y = Result /
                    GROUP = Measurement_Type;


            ENDLAYOUT;

        ENDGRAPH;
    END;
RUN;

/* Grafik - Erstellung / Ausführung */
/*****/
PROC SGRENDER    DATA          = KSFE.GTL_Example
                   TEMPLATE     = work.KSFE_GLT_Plot;
RUN;
/*****/

ODS LAYOUT END; ODS PDF CLOSE; ODS LISTING;
```

1.2 PDF Output: Step 1



2 Step 10: Summary

2.1 SAS Programm: Step 10

Das folgende Programm enthält die Inhalte von Step 1 plus weitere **Plot Statements** und **Elemente von SAS GTL**:

```
LIBNAME KSFE          "D:\KSFE_GTL\Daten";
%LET Result_Path     = D:\KSFE_GTL\PDF;

OPTIONS      NODATE NONUMBER ORIENTATION=portrait PAPERSIZE='A4'
             NOCENTER TOPMARGIN=0.7CM  BOTTOMMARGIN=1.0CM
             LEFTMARGIN=2.0CM  RIGHTMARGIN=1.0CM;
GOPTIONS    RESET=ALL  DEVICE=winprtc  ROTATE=PORTRAIT;

ODS LISTING CLOSE;
ODS PDF FILE = "&Result_Path.\Step_10.pdf" NOTOC STYLE=PRINTER;
ODS LAYOUT START;
ODS GRAPHICS ON / BORDER=off          WIDTH=17.9cm  HEIGHT=14.9cm;
ODS REGION      X=0.0cm  Y=0.5cm  WIDTH=18.0cm  HEIGHT=15.0cm;
```

```
DATA _NULL_ ;
SET KSFE.GTL_Statistics;
CALL SYMPUT ("Y_Result_Limit_High", Result_Limit_High);
CALL SYMPUT ("Y_Result_Limit_Low", Result_Limit_Low);
CALL SYMPUT
("X_Stability_Limit", STRIP(PUT(Stability_Limit, 11.0)));
```

```
RUN;
```

```
/******
```

```
PROC TEMPLATE;
```

```
DEFINE STATGRAPH work.KSFE_GLT_Plot;
```

```
BEGINGRAPH / BORDER = YES BORDERATTRS = (color=brown) ;
```

```
DYNAMIC Title_Plot Y_Label Y_Min Y_Max ;
```

```
ENTRYTITLE
```

```
HALIGN = LEFT
```

```
TEXTATTRS = (family="Arial" weight=bold size=14pt)
```

```
Title_Plot ;
```

```
/* for Symbols */
```

```
/******
```

```
DISCRETEATTRMAP NAME = "Symbol_Map"
```

```
VALUE "Initial Measurement" /
```

```
MARKERATTRS = (color=blue symbol=diamondfilled);
```

```
VALUE "Longterm Measurement" /
```

```
MARKERATTRS = (color=orange symbol=circlefilled);
```

```
ENDDISCRETEATTRMAP;
```

```
DISCRETEATTRVAR ATTRVAR = Symbols_Measurement_Type
```

```
ATTRMAP = "Symbol_Map"
```

```
VAR = Measurement_Type;
```

```
LAYOUT OVERLAY /
```

```
/* Y-Axis */
```

```
/******
```

```
YAXISOPTS = (LINEAROPTS = (viewmin=Y_Min viewmax=Y_Max)
```

```
LABEL = Y_Label
```

```
LABELATTRS = (family="Arial" size=14pt)
```

```
TICKVALUEATTRS = (family="Arial" color=grey)
```

```
OFFSETMIN = 0.05
```

```
OFFSETMAX = 0.05)
```

```

/* X-Axis */
/******/
XAXISOPTS = (LABELATTRS = (family="Arial" size=13pt)
             TICKVALUEATTRS = (family="Arial" color=grey)
             OFFSETMIN = 0.04
             OFFSETMAX = 0.04 ) ;

/* Blue Band - Data Range */
/******/
BANDPLOT X = Hours LIMITUPPER = &Y_Result_Limit_High.
          LIMITLOWER = &Y_Result_Limit_Low. /
          DATATRANSPARENCY = 0.95
          FILLATTRS = (color=blue)
          EXTEND = true;

REFERENCELINE Y = &Y_Result_Limit_High. /
LINEATTRS = (pattern=solid thickness=1 color=darkblue)
CURVELABEL = "Upper Limit"
CURVELABELLOCATION = inside
CURVELABELPOSITION = min
CURVELABELATTRS =
    (family="Arial" size=8pt color=darkblue);

REFERENCELINE Y = &Y_Result_Limit_Low. /
LINEATTRS = (pattern=solid thickness=1 color=darkblue)
CURVELABEL = "Lower Limit"
CURVELABELLOCATION = inside
CURVELABELPOSITION = min
CURVELABELATTRS =
    (family="Arial" size=8pt color=darkblue);

/* Green Line - Stability Limit */
/******/
REFERENCELINE X = &X_Stability_Limit. /
LINEATTRS = (pattern=solid thickness=1 color=green)
CURVELABEL =
    "Estimated Stability = &X_Stability_Limit. Hours"
CURVELABELATTRS =
    (family="Arial" size=10pt color=green)
CURVELABELPOSITION = max;

/* Grey Regression Line */
/******/
REGRESSIONPLOT X = Hours Y = Result /
LINEATTRS = (thickness=2 color=grey)
CLM = "Konf_Limits_Name"
NAME = "Regression Line"
LEGENDLABEL = "Linear Regression";

```

```
/* Red Confidence Limits */
/*****/
MODELBAND "Konf_Limits_Name" /
  DISPLAY = (outline)
  OUTLINEATTRS = (pattern=shortdash thickness=1 color=red)
  NAME = "Limits"
  LEGENDLABEL = "Confidence Limits";

/* Blue/Yellow Data Points - Measured Values */
/*****/
SCATTERPLOT X = Hours Y =Result /
  GROUP = Symbols_Measurement_Type
  MARKERATTRS = (weight=bold size=4pt)
  DATATRANSARENCY = 0.6
  NAME = "Measured Points";

/* Legend Bottom Right */
/*****/
DISCRETELEGEND "Measured Points"
               "Regression Line"
               "Limits" /
  BORDER = false
  ORDER = columnmajor
  VALIGN = bottom
  HALIGN = right
  VALUEATTRS = (family="Arial" size=10pt color=grey);

  ENDLAYOUT;

  ENDGRAPH;

  END;

  RUN;

/*****/

PROC SGRENDER DATA = KSFE.GTL_Example
              TEMPLATE = work.KSFE_GLT_Plot;


  DYNAMIC Title_Plot = "Summary"
          Y_Label = "Concentration [ µg/L ]"
          Y_Min = 182.5
          Y_Max = 195 ;

  RUN;

/*****/

ODS LAYOUT END; ODS PDF CLOSE; ODS LISTING;
```

2.2 PDF Output: Step 10



3 Anmerkung zur Programm-Ausführung

Um die Programme ausführen zu können, müssen die Verzeichnisse „D:\KSFE_GTL\Daten“ und „D:\KSFE_GTL\PDF“ vorhanden sein. Es sind auch andere Verzeichnisse möglich, wenn man Libname in Zeile 1 und Ergebnisfad-Name in Zeile 2 des Programmcodes entsprechend anpasst.

Im Daten-Verzeichnis befinden sich die Datasets „GTL_Example“ und „GTL_Statistics“ (siehe Kapitel 4).

4 Verwendete Datasets

4.1 Dataset „GTL_Example“

ID	Sample_ID	Hours	Result	Unit	Measurement_Type	Result_Type
1	Pool	0	185,5	µg/L	Initial Measurement	Concentration
2	Pool	0,25	187,5	µg/L	Initial Measurement	Concentration
3	Pool	0,5	186,5	µg/L	Initial Measurement	Concentration
4	Pool	0,75	187	µg/L	Initial Measurement	Concentration
5	Pool	12	185,8	µg/L	Longterm Measurement	Concentration
6	Pool	24	187,1	µg/L	Longterm Measurement	Concentration
7	Pool	36	185,8	µg/L	Longterm Measurement	Concentration
8	Pool	48	186,8	µg/L	Longterm Measurement	Concentration
9	Pool	60	185	µg/L	Longterm Measurement	Concentration
10	Pool	72	187,9	µg/L	Longterm Measurement	Concentration
11	Pool	84	186,1	µg/L	Longterm Measurement	Concentration
12	Pool	96	188,1	µg/L	Longterm Measurement	Concentration
13	Pool	108	187,3	µg/L	Longterm Measurement	Concentration
14	Pool	120	186,4	µg/L	Longterm Measurement	Concentration
15	Pool	132	187,8	µg/L	Longterm Measurement	Concentration
16	Pool	144	186,7	µg/L	Longterm Measurement	Concentration
17	Pool	156	185,9	µg/L	Longterm Measurement	Concentration
18	Pool	168	188,4	µg/L	Longterm Measurement	Concentration
19	Pool	180	186,7	µg/L	Longterm Measurement	Concentration
20	Pool	192	188,2	µg/L	Longterm Measurement	Concentration
21	Pool	204	186,2	µg/L	Longterm Measurement	Concentration
22	Pool	216	188,1	µg/L	Longterm Measurement	Concentration
23	Pool	228	188,8	µg/L	Longterm Measurement	Concentration
24	Pool	240	186,8	µg/L	Longterm Measurement	Concentration
25	Pool	252	187,7	µg/L	Longterm Measurement	Concentration
26	Pool	264	188,8	µg/L	Longterm Measurement	Concentration
27	Pool	276	186,8	µg/L	Longterm Measurement	Concentration
28	Pool	288	188,7	µg/L	Longterm Measurement	Concentration
29	Pool	300	188,4	µg/L	Longterm Measurement	Concentration
30	Pool	312	187,2	µg/L	Longterm Measurement	Concentration
31	Pool	324	188,6	µg/L	Longterm Measurement	Concentration
32	Pool	336	189,2	µg/L	Longterm Measurement	Concentration
33	Pool	348	189,7	µg/L	Longterm Measurement	Concentration
34	Pool	360	187,9	µg/L	Longterm Measurement	Concentration
35	Pool	372	188,9	µg/L	Longterm Measurement	Concentration
36	Pool	384	189,9	µg/L	Longterm Measurement	Concentration
37	Pool	396	187,5	µg/L	Longterm Measurement	Concentration
38	Pool	408	189,5	µg/L	Longterm Measurement	Concentration
39	Pool	420	190,7	µg/L	Longterm Measurement	Concentration
40	Pool	432	190,3	µg/L	Longterm Measurement	Concentration
41	Pool	444	188,4	µg/L	Longterm Measurement	Concentration
42	Pool	456	190,9	µg/L	Longterm Measurement	Concentration
43	Pool	468	188,4	µg/L	Longterm Measurement	Concentration
44	Pool	480	191,2	µg/L	Longterm Measurement	Concentration

4.2 Dataset „GTL_Statistics“

ID	Intercept	Slope	Result_Limit_High	Result_Limit_Low	Stability_Limit
1	186,14534975	0,0075904664	189	184	328

5 Anmerkung zu Step 5

Die in Step 5 erwähnten Statements zu „Line (Slope/Intercept), Connection of Data Points“ sind in Step 10 nicht enthalten, da Step 5 als Alternative zu Step 4 möglich ist, wenn man vorher entsprechende Berechnungen programmiert hat.

Für diese beiden Grafik-Elemente werden folgende GTL Plot Statements verwendet:

- **Line (Slope/Intercept):**

```
LINEPARM X= 0 Y= ... SLOPE= ... ;
```

Y-Achsenabschnitt (Intercept) und Steigung (Slope) der Geraden müssen vorher berechnet werden und können als Variable an das GTL Template übergeben werden.

- **Connection of Data Points:**

```
SERIESPLOT X= ... Y= ... ;
```

X und Y werden Dataset-Variablen zugewiesen, die schon in den Daten vorliegen bzw. vor der Grafik-Erstellung berechnet werden.

6 Informationen

- SAS GTL ist Teil von **SAS Base** ab SAS 9.3
- Ausführliche **Hilfe** in „SAS Help und Documentation“:
SAS Products -> Base SAS -> ODS Graphics ->
SAS 9.3 Graph Template Language: Reference
- **ODS Graphics Designer** im SAS im „Tools“ - Menü
- **Flexibilisierung** möglich durch : Dataset (z. B. Hybrid Dataset),
Dynamic Variablen, Macro-Variablen, IF-ELSE, %IF-%ELSE, Functions
- **Komplexität** möglich durch:
10 verschiedene Layouts (z.B. Layout Lattice), 35 Plot Statements, Text, Draw...

7 Fazit

SAS GTL ist ein mächtiges Werkzeug bestehend aus einem gut strukturierten und durchdachten „Baukasten-System“. Beispiele in der Hilfe und im ODS Graphics Designer sind eine gute Orientierung für die eigene Programmierung.